

Estimating Impulse Response Functions in Stata

David Schenck

Stata

January 26, 2024


Portugal Stata Research Symposium

Outline

- Impulse response functions
- Impulse response functions in Stata
- The local projection estimator
- Instruments in impulse response estimation

Introduction

- Impulse response functions trace out the path of an outcome after an unexpected disturbance.
- Key object of interest in time-series econometrics.
- Teaser: one model's predictions of the paths of inflation and the output gap after a monetary policy shock:



Graphs by ifname, impulse variable, and response variable

The setting

- Collection of time–series variables driven by unobserved disturbances
 - Price and quantity driven by supply and demand shocks
 - Output, inflation, and interest rates driven by aggregate demand, aggregate supply, and monetary policy shocks
- Question: What effect does an unexpected, one-time increase in a disturbance have on the outcomes of interest?
- Policy questions:
 - How does inflation respond to an oil price increase?
 - How does GDP respond to an interest rate increase?
 - How does employment respond to a government spending increase?

Answering policy questions

- Answers are inherently dynamic: outcome variables might only respond to a disturbance after a lag.
- Effects one year or five years out might differ from effects on impact.
- Questions of causal inference at the forefront: how to disentangle causes from observed time series?
- Instruments and exogenous policy actions are only occasionally available.

An abstract definition

- Let y_t be an outcome of interest
- Let x_t be a disturbance
- We are interested in the time path of the variable in periods following a shock, relative to no shock:

$$IRF(h) = E[y_{t+h}|x_t = 1] - E[y_{t+h}|x_t = 0]$$

A model-centric definition

- Define and estimate a model for one or more outcomes \mathbf{y}_t in terms of past values, observed exogenous variables, and unobserved disturbances.
- The simplest possible model:

$$y_t = \alpha y_{t-1} + e_t$$

- Then the impulse–response function after an unexpected 1-unit increase in e_t is: $1, \alpha, \alpha^2, \dots$
- Further lags create more interesting short-run dynamics
- Eventually, the impulse–response function converges to 0

Impulse responses in multiple-equation models

- With a collection of variables, identification issues become crucial
- Two variables, one lag, two shocks:

$$y_{1t} = a_{11}y_{1,t-1} + a_{12}y_{2,t-1} + u_{1t}$$

$$y_{2t} = a_{21}y_{1,t-1} + a_{22}y_{2,t-1} + u_{2t}$$

- The residuals (u_{1t}, u_{2t}) may be correlated
- Wish to decompose into economically meaningful shocks

$$\begin{pmatrix} u_{1t} \\ u_{2t} \end{pmatrix} = \begin{pmatrix} b_{11} & b_{12} \\ b_{21} & b_{22} \end{pmatrix} \begin{pmatrix} e_{1t} \\ e_{2t} \end{pmatrix}$$

where (e_{1t}, e_{2t}) are orthogonal

- Triangular decompositions
- More creative decompositions

Impulse responses in other models

- Autoregressive moving average models,

$$y_t = a_1 y_{t-1} + \cdots + a_p y_{t-p} + e_t + \psi_1 e_{t-1} + \cdots + \psi_q e_{t-q}$$

- Structural vector autoregression models,

$$\mathbf{y}_t = \mathbf{A}_1 \mathbf{y}_{t-1} + \cdots + \mathbf{A}_p \mathbf{y}_{t-p} + \mathbf{B} e_t$$

- Vector error correction models,

$$\Delta \mathbf{y}_t = \boldsymbol{\Gamma} \mathbf{y}_{t-1} + \mathbf{A}_1 \Delta \mathbf{y}_{t-1} + \cdots + \mathbf{A}_p \Delta \mathbf{y}_{t-p} + \mathbf{u}_t$$

- DSGE models,

$$\begin{aligned} \mathbf{y}_t &= \mathbf{G}(\boldsymbol{\theta}) \mathbf{x}_t \\ \mathbf{x}_{t+1} &= \mathbf{H}(\boldsymbol{\theta}) \mathbf{x}_t + \mathbf{M}(\boldsymbol{\theta}) e_{t+1} \end{aligned}$$

Impulse responses as moving-average coefficients I

- Regardless of the model, we perform two steps: use the model to obtain moving-average coefficients, and have some method for disentangling residuals into structural shocks
- Consider a vector autoregression:

$$\mathbf{y}_t = \mathbf{A}_1\mathbf{y}_{t-1} + \cdots + \mathbf{A}_p\mathbf{y}_{t-p} + \mathbf{u}_t$$

- Invert to write \mathbf{y}_t as a function of current and past shocks

$$\mathbf{y}_t = \mathbf{u}_t + \Phi_1\mathbf{u}_{t-1} + \Phi_2\mathbf{u}_{t-2} + \dots$$

The Φ_i are the simple impulse–response coefficients

- Combine with a theory for how residuals map into shocks

$$\mathbf{u}_t = \mathbf{B}\mathbf{e}_t$$

Impulse responses as moving-average coefficients II

- Then we can write

$$\mathbf{y}_t = \mathbf{B}\mathbf{e}_t + \boldsymbol{\Phi}_1\mathbf{B}\mathbf{e}_{t-1} + \boldsymbol{\Phi}_2\mathbf{B}\mathbf{e}_{t-2} + \dots$$

which are “structural” impulse–response functions

- Punchline: need to identify $(\mathbf{B}, \boldsymbol{\Phi}_i)$
- The $\boldsymbol{\Phi}_i$ can be consistently estimated (assuming correct specification) solely from the reduced form
- IRF identification problems are problems with the impact matrix \mathbf{B}

IRF identification as a simultaneous-equations problem

- Alternate way to think about the identification problem.
- A two-equation structural VAR is equivalent to

$$y_{1t} = b_{12}y_{2t} + \cdots + e_{1t}$$

$$y_{2t} = b_{21}y_{1t} + \cdots + e_{2t}$$

- No amount of information on (y_{1t}, y_{2t}) alone can identify both b_{12} and b_{21}
- Situation extrapolates to larger models with more equations.
- Structural models make assumptions on some of these coefficients (say, setting them equal to 0) to estimate the remaining ones.

Impulse response functions in Stata

The irf suite of commands

- Stata has a suite of commands to create, manage, graph, and tabulate impulse responses
 - `irf set`
 - `irf create`
 - `irf describe`
 - `irf table` and `irf graph`
- This suite is available after estimating many time-series models
 - `arima` and `arfima`
 - `var` and `svar`
 - `vec`
 - `dsge` and `dsge1`
 - `lpirf`
- Bayesian IRFs are available as well after estimating
 - `bayes: var`
 - `bayes: dsge` and `bayes: dsge1`

Data and questions

```
. use usmacro3
```

```
. describe
```

```
Contains data from dta/usmacro3.dta
```

```
Observations:      312
```

```
Variables:          5
```

```
24 Jan 2024 15:48
```


| Variable name | Storage type | Display format | Value label | Variable label |
|---------------|--------------|----------------|-------------|------------------------|
| date | float | %tq | | Quarterly date |
| fedfunds | double | %10.0g | | Federal funds rate |
| inflation | float | %9.0g | | Inflation |
| ogap | float | %9.0g | | GDP gap |
| netoil | float | %9.0g | | Net oil price increase |

```
Sorted by: date
```

```
Note: Dataset has changed since last saved.
```

Data and questions

```
. tsline inflation ogap fedfunds, yline(0) legend(rows(1))
```



Estimating a VAR

```
. quietly var inflation ogap fedfunds, lag(1/4)
. // (setup omitted)
. collect preview
```

| | Inflation | | GDP gap | | Federal funds rate | |
|--------------|-----------|-----------|---------|-----------|--------------------|-----------|
| | Coef. | Std. Err. | Coef. | Std. Err. | Coef. | Std. Err. |
| L.inflation | 1.15 | (0.07) | -0.09 | (0.09) | -0.08 | (0.09) |
| L2.inflation | -0.22 | (0.11) | 0.12 | (0.13) | 0.32 | (0.14) |
| L3.inflation | 0.10 | (0.11) | -0.19 | (0.13) | -0.16 | (0.14) |
| L4.inflation | -0.10 | (0.07) | 0.13 | (0.08) | 0.03 | (0.09) |
| L.ogap | 0.09 | (0.06) | 1.20 | (0.07) | 0.38 | (0.07) |
| L2.ogap | -0.11 | (0.08) | -0.10 | (0.10) | -0.21 | (0.11) |
| L3.ogap | 0.01 | (0.08) | -0.18 | (0.10) | -0.10 | (0.11) |
| L4.ogap | 0.03 | (0.06) | 0.01 | (0.07) | -0.02 | (0.07) |
| L.fedfunds | 0.21 | (0.05) | 0.11 | (0.07) | 1.14 | (0.07) |
| L2.fedfunds | -0.16 | (0.08) | -0.38 | (0.10) | -0.52 | (0.10) |
| L3.fedfunds | 0.16 | (0.08) | 0.38 | (0.10) | 0.47 | (0.10) |
| L4.fedfunds | -0.19 | (0.05) | -0.13 | (0.07) | -0.19 | (0.07) |
| Intercept | 0.20 | (0.08) | 0.16 | (0.10) | 0.16 | (0.11) |

(with an assist from collect)

Computing IRFs

- `irf set filename` sets a destination file for IRF results. It can hold results from multiple models, facilitating model comparison.
- `irf create irfname` estimates IRFs using the most recently estimated time-series model and sends the results to the file specified in `irf set`.


```
. quietly var inflation ogap fedfunds, lag(1/8)
. irf set modelirfs.irf, replace
(file modelirfs.irf created)
(file modelirfs.irf now active)
. irf create varmodel, step(20)
(file modelirfs.irf updated)
```

Graphing IRFs

- `irf graph irftype` graphs IRFs and related statistics
- Many models produce several kinds of IRFs: simple, orthogonalized, and/or structural; along with dynamic multipliers (responses to impulses to exogenous variables)
- We can look at the simple IRFs:

Graphing IRFs II

```
. irf graph irf, yline(0) xlabel(0(4)20)
```



Graphs by irfname, impulse variable, and response variable

Orthogonalized impulse responses I

- The simple IRFs shown above are responses to impulses to the VAR residuals, or forecast errors.
- They have statistical meaning, but not economic meaning because the VAR residuals are correlated.
- For economic interpretation, we need shocks that are orthogonal, so that it is meaningful to change one without changing the others.
- Consider again the VAR

$$\mathbf{y}_t = \mathbf{A}_1 \mathbf{y}_{t-1} + \cdots + \mathbf{A}_p \mathbf{y}_{t-p} + \mathbf{u}_t$$

$$\mathbf{u}_t = \mathbf{B} \mathbf{e}_t$$

- We can always calculate the covariance matrix of the VAR residuals \mathbf{U}

$$\boldsymbol{\Sigma}_u = \frac{1}{T} \mathbf{U}' \mathbf{U}$$

Orthogonalized impulse responses II

- The covariance of VAR residuals is related to the impact matrix \mathbf{B}

$$\Sigma_{\mathbf{u}} = \mathbf{B}\mathbf{B}'$$


- $\Sigma_{\mathbf{u}}$ is symmetric, positive definite, with $k(k + 1)/2$ unique elements
- \mathbf{B} is, in principle, completely unrestricted with k^2 elements
- Orthogonalization solves this identification problem by assuming \mathbf{B} is lower-triangular.
- Orthogonalization produces statistically uncorrelated shocks by construction.
- Hence it is sensible to consider a disturbance to one shock, holding the others constant.
- But still no guarantee that these shocks are economically meaningful

Orthogonalized IRFs in Stata

- Orthogonalization is such a common identification scheme that Stata does it automatically when creating IRFs after `var`.
- We can graph them with `irf graph oirf`

Orthogonalized IRFs

```
. irf graph oirf, yline(0)
```



Graphs by irfname, impulse variable, and response variable



Identification in structural VARs I

- Orthogonalization consisted of the model

$$\mathbf{y}_t = \mathbf{A}_1 \mathbf{y}_{t-1} + \cdots + \mathbf{A}_p \mathbf{y}_{t-p} + \mathbf{u}_t$$

$$\mathbf{u}_t = \mathbf{B} \mathbf{e}_t$$

B lower triangular

- Orthogonalization is equivalent to giving the shocks a recursive structure
- A general structural VAR with short-run restrictions takes the form

$$\mathbf{y}_t = \mathbf{A}_1 \mathbf{y}_{t-1} + \cdots + \mathbf{A}_p \mathbf{y}_{t-p} + \mathbf{u}_t$$

$$\mathbf{A} \mathbf{u}_t = \mathbf{B} \mathbf{e}_t$$

where some terms in (\mathbf{A}, \mathbf{B}) are restricted on the basis of theory.

Structural VARs and structural IRFs I

- svar estimates structural VARs with short-run or long-run restrictions
- Stata's matrices are used to define constraints on the matrices (**A**, **B**)
- Example: Orthogonalization sets $\mathbf{A} = \mathbf{I}$ and sets the upper-triangular portion of **B** to zero. The remaining parameters in **B** are estimated. Setting this up in Stata involves:

```
. matrix A = I(3)
. matrix B = (., 0, 0 \ ., ., 0 \ ., ., .)
. matlist A
```

| | c1 | c2 | c3 |
|----|----|----|----|
| r1 | 1 | | |
| r2 | 0 | 1 | |
| r3 | 0 | 0 | 1 |

```
. matlist B
```

| | c1 | c2 | c3 |
|----|----|----|----|
| r1 | . | 0 | 0 |
| r2 | . | . | 0 |
| r3 | . | . | . |

Structural VARs and structural IRFs II

● Estimation:

```
. svar inflation ogap fedfunds, aeq(A) beq(B) nocnsreport nolog  
Estimating short-run parameters
```

```
Structural vector autoregression
```

```
Sample: 1956q1 thru 2010q4
```

```
Number of obs = 220
```


```
Exactly identified model
```

```
Log likelihood = -723.8898
```

| | Coefficient | Std. err. | z | P> z | [95% conf. interval] | |
|-----|-------------|---------------|-------|-------|----------------------|----------|
| /A | | | | | | |
| 1_1 | 1 | (constrained) | | | | |
| 2_1 | 0 | (constrained) | | | | |
| 3_1 | 0 | (constrained) | | | | |
| 1_2 | 0 | (constrained) | | | | |
| 2_2 | 1 | (constrained) | | | | |
| 3_2 | 0 | (constrained) | | | | |
| 1_3 | 0 | (constrained) | | | | |
| 2_3 | 0 | (constrained) | | | | |
| 3_3 | 1 | (constrained) | | | | |
| /B | | | | | | |
| 1_1 | .6294081 | .0300059 | 20.98 | 0.000 | .5705977 | .6882185 |
| 2_1 | .0784342 | .0532413 | 1.47 | 0.141 | -.0259169 | .1827853 |
| 3_1 | .2313289 | .0550465 | 4.20 | 0.000 | .1234397 | .3392181 |
| 1_2 | 0 | (constrained) | | | | |
| 2_2 | .7877468 | .0375544 | 20.98 | 0.000 | .7141416 | .861352 |
| 3_2 | .2260354 | .052843 | 4.28 | 0.000 | .122465 | .3296057 |
| 1_3 | 0 | (constrained) | | | | |
| 2_3 | 0 | (constrained) | | | | |
| 3_3 | .7673186 | .0365805 | 20.98 | 0.000 | .6956222 | .839015 |

Structural VARs and structural IRFs III

```
. irf create svarmodel, step(20)  
(file modelirfs.irf updated)  
  
. irf graph sirf, irf(svarmodel) yline(0) xlabel(0(4)20)
```



Graphs by irfname, impulse variable, and response variable

Structural VARs with non-recursive constraints I

- The constraints in `svar` need not be recursive
- Setup:

```
. matrix A = (., 0, 0 \ 0, ., 0 \ 0, 0, .)
. matrix B = (1, ., 0 \ 0, 1, . \ ., 0, 1)
. matlist A
```

| | c1 | c2 | c3 |
|----|----|----|----|
| r1 | . | | |
| r2 | 0 | . | |
| r3 | 0 | 0 | . |

```
. matlist B
```

| | c1 | c2 | c3 |
|----|----|----|----|
| r1 | 1 | . | 0 |
| r2 | 0 | 1 | . |
| r3 | . | 0 | 1 |

Structural VARs with non-recursive constraints II

Estimation:

```
. svar inflation ogap fedfunds, aeq(A) beq(B) nocnsreport nolog
Estimating short-run parameters
Structural vector autoregression
Sample: 1956q1 thru 2010q4           Number of obs   =       220
Exactly identified model             Log likelihood   =  -723.8898
```

| | Coefficient | Std. err. | z | P> z | [95% conf. interval] | |
|-----|-------------|---------------|------|-------|----------------------|----------|
| /A | | | | | | |
| 1_1 | 5.456983 | 1.311046 | 4.16 | 0.000 | 2.88738 | 8.026585 |
| 2_1 | 0 | (constrained) | | | | |
| 3_1 | 0 | (constrained) | | | | |
| 1_2 | 0 | (constrained) | | | | |
| 2_2 | 12.19719 | 8.362181 | 1.46 | 0.145 | -4.192381 | 28.58676 |
| 3_2 | 0 | (constrained) | | | | |
| 1_3 | 0 | (constrained) | | | | |
| 2_3 | 0 | (constrained) | | | | |
| 3_3 | 4.013121 | .8971846 | 4.47 | 0.000 | 2.254671 | 5.77157 |
| /B | | | | | | |
| 1_1 | 1 | (constrained) | | | | |
| 2_1 | 0 | (constrained) | | | | |
| 3_1 | 3.188577 | .7303245 | 4.37 | 0.000 | 1.757167 | 4.619987 |
| 1_2 | 3.285871 | .8124037 | 4.04 | 0.000 | 1.69359 | 4.878153 |
| 2_2 | 1 | (constrained) | | | | |
| 3_2 | 0 | (constrained) | | | | |
| 1_3 | 0 | (constrained) | | | | |
| 2_3 | 9.603887 | 6.606896 | 1.45 | 0.146 | -3.345392 | 22.55317 |
| 3_3 | 1 | (constrained) | | | | |

Recap

- We've seen how to set up and estimate IRFs in Stata
- We've seen a few methods for identifying the impact matrix **B**
- Next: the local projection estimator

The local projection estimator

Local projections

- In a VAR, we estimate the $(\mathbf{A}_1, \dots, \mathbf{A}_p)$ lag coefficients, then invert them to obtain the $(\boldsymbol{\Phi}_1, \boldsymbol{\Phi}_2, \dots)$ moving-average coefficients.
- Local projections (Jorda 2005) estimate the MA coefficients directly.
- Moving-average coefficients computed on the basis of “long” regressions

$$\mathbf{y}_{t+h} = \boldsymbol{\Phi}_{h+1} \mathbf{y}_{t-1} + \mathbf{u}_{t+h}$$

- Popularly combined with exogenous variables:

$$y_{t+h} = \beta_h x_t + \text{controls} + u_{t+h}$$

with x_t an observed or constructed, exogenous impulse of interest.

LP in the context of VAR

- Both VAR and LP produce estimates of the moving average coefficients
- LPs provide a simpler and more convenient way to impose cross-equation restrictions and perform tests directly on the moving average coefficients
- But identification of structural IRFs still requires an estimate of the impact matrix, which LPs do not necessarily help with.
- ... hence the rising popularity of constructed exogenous regressors in VAR/LP studies.

LP estimation

- LPs estimate the h moving average coefficients directly.
- Most LPs also include lags of the dependent variable as controls.
- With p lags and h impulse–response horizons, we lose $p + h$ observations: p in the beginning of the sample and h at the end.
- Hence small-sample issues are magnified relative to a VAR.
- Stata estimates the full set of IRFs, at all horizons for all variables, jointly.
- This allows easy tests of coefficients across horizons and across variables

LP estimation with `lpirf`

```
. lpirf ogap, lags(1/4)
```

```
Local-projection impulse-responses
```

```
Sample: 1955q3 thru 2009q1
```

```
Number of obs      = 215  
Number of impulses = 1  
Number of responses = 1  
Number of controls = 3
```

| | IRF | | | | | |
|------|-------------|-----------|-------|-------|----------------------|----------|
| | coefficient | Std. err. | z | P> z | [95% conf. interval] | |
| ogap | | | | | | |
| F1. | 1.222087 | .0685199 | 17.84 | 0.000 | 1.087791 | 1.356384 |
| F2. | 1.354971 | .1081866 | 12.52 | 0.000 | 1.142929 | 1.567013 |
| F3. | 1.301492 | .1425385 | 9.13 | 0.000 | 1.022122 | 1.580863 |
| F4. | 1.218761 | .1682287 | 7.24 | 0.000 | .8890384 | 1.548483 |
| F5. | 1.026617 | .1881264 | 5.46 | 0.000 | .6578963 | 1.395338 |
| F6. | .9274278 | .2011277 | 4.61 | 0.000 | .5332248 | 1.321631 |
| F7. | .8099336 | .2112339 | 3.83 | 0.000 | .3959227 | 1.223944 |
| F8. | .5991371 | .2186316 | 2.74 | 0.006 | .1706271 | 1.027647 |


```
Impulses: ogap
```

```
Responses: ogap
```

```
Controls: L2.ogap L3.ogap L4.ogap
```

LP IRF graphs

```
. irf create lp_ogap  
(file modelirfs.irf updated)  
. irf graph irf, irf(lp_ogap) yline(0)
```



LPs and VARs: further comparisons


- LPs and VARs both estimate moving-average coefficients
- LPs and VARs both require additional identifying assumptions to make causal statements about their IRFs
- In principle, any identification strategy used in a VAR can also be used in an LP
- Stata implements orthogonalized IRFs for LPs

IRF comparison I

- We can estimate LP IRFs for the same model specification as in the VAR before, graphing the orthogonalized IRFs

```
. quietly lpirf inflation ogap fedfunds, lags(1/8) step(20)
. irf create lpmodel
(file modelirfs.irf updated)
. irf graph oirf, irf(lpmodel) yline(0)
```


IRF comparison II



Graphs by irfname, impulse variable, and response variable

IRF comparison III

```
. irf graph oirf, irf(varmodel lpmodel) impulse(fedfunds)
```




Graphs by irfname, impulse variable, and response variable

LPs with exogenous variables I

- `lpirf` allows for exogenous variables and computes dynamic multipliers for them.
- Convention: dynamic multipliers computed for any t -dated variable in the set of controls.
- The dataset I have been using has a constructed measure of oil price shocks, the *net oil price increase* (Hamilton 1996, 2003)
- The net oil price increase is the difference between the price of oil in period t and its maximum in the previous $t - \ell$ quarters.
- Captures persistent movements in the price of oil.

LPs with exogenous variables II



LPs with exogenous variables III

- I estimate LPs for the effect of a net oil price increase on outcome variables, modelling the net oil price increase as exogenous.
- For outcomes inflation, output gap, and interest rate, run


$$y_{t+h} = \beta_h \text{netoil}_t + \mathbf{z}'\boldsymbol{\gamma} + u_{t+h}$$

where controls \mathbf{z} include the first 8 lags of all outcome variables and of the net oil price increase

```
. quietly lpirf inflation ogap fedfunds, exog(L(0/8).netoil) ///  
>       lags(1/8) step(17)  
  
. irf create lpoil_full  
(file modelirfs.irf updated)  
  
. irf graph dm, irf(lpoil_full) impulse(netoil) yline(0) xlabel(0(4)16)
```

- The resulting impulse responses for a 1% net oil price increase are:

LPs with exogenous variables IV



Graphs by irfname, impulse variable, and response variable


LPs with exogenous variables V

- For comparison, recompute the LPs for the sample through 1985q4,

```
. quietly lpirf inflation ogap fedfunds if tin(, 1985q4), ///  
>       exog(L(0/8).netoil) lags(1/8) step(17)  
  
. irf create lpoil_early  
(file modelirfs.irf updated)  
  
. irf graph dm, irf(lpoil_full lpoil_early) impulse(netoil) ///  
>       yline(0) xlabel(0(4)16)
```

- Comparing the resulting IRFs:

LPs with exogenous variables VI



Graphs by irfname, impulse variable, and response variable

Summary

- Local projections provide an alternate method for constructing impulse response functions
- LP-IRFs are computed directly, simultaneously, jointly, allowing for easier tests of coefficients
- LPs do not solve identification problems, but can be combined with identification strategies
- Being a regression-based method, clear parallels with and extensions to other regression-based methods

Instruments in impulse response estimation

Instruments in micro

- Consider estimating a coefficient in a regression

$$y_t = \beta x_t + e_t$$

- When $\text{cov}(x_t, e_t) \neq 0$ we say x_t is endogenous and we know OLS estimation is inconsistent
- Solution: bring in an instrument z_t with the properties

$$\text{cov}(z_t, x_t) \neq 0 \quad \text{(relevance)}$$

$$\text{cov}(z_t, e_t) = 0 \quad \text{(exclusion)}$$

- Macroeconometrics has a notion of an instrument as well

Instruments in macro I

- Macro systems are driven by unobserved shocks
- An instrument is an observed variable the researcher believes to satisfy

$$\text{cov}(z_t, e_{1t}) \neq 0 \quad \text{(relevance)}$$

$$\text{cov}(z_t, e_{jt}) = 0 \quad \forall j \neq 1 \quad \text{(exclusion)}$$

(Stock and Watson 2012; Gertler and Karadi 2015; many others)

- An instrument thus satisfies

$$z_t = \gamma e_{1,t} + w_t$$

where $\gamma \neq 0$ ensures relevance, there is no feedback from other shocks, and w_t allows the instrument to be noisy

Instruments in macro II

- Consider a VAR augmented with an instrument:

$$\mathbf{y}_t = \mathbf{A}(\ell)\mathbf{y}_{t-1} + \mathbf{u}_t$$

$$\mathbf{u}_t = \mathbf{B}\mathbf{e}_t$$

$$z_t = \gamma e_{1,t} + w_t$$

Then we can write

$$\begin{pmatrix} \mathbf{y}_t \\ z_t \end{pmatrix} = \begin{pmatrix} \mathbf{A}_y(\ell) & \mathbf{0} \\ \mathbf{A}_{zy}(\ell) & \mathbf{A}_z(\ell) \end{pmatrix} \begin{pmatrix} \mathbf{y}_{t-1} \\ z_{t-1} \end{pmatrix} + \begin{pmatrix} \mathbf{B}_1 & \mathbf{B}_2 & \mathbf{0} \\ \gamma & \mathbf{0} & \sigma_w \end{pmatrix} \begin{pmatrix} e_{1t} \\ \mathbf{e}_{jt} \\ w_t \end{pmatrix}$$

- Thus, an SVAR-IV is just a large SVAR (Angelini and Fanelli 2019)
- \mathbf{B}_1 contains impact effects for the shock being instrumented
- \mathbf{B}_2 contains columns of impact effects for shocks without instruments
- The $\mathbf{0}$ blocks are implied by the properties of the instrument

SVAR-IV estimation of oil shock IRFs I

- We estimate a VAR in inflation, the output gap, and interest rates, using net oil price shocks as an instrument for “inflation shocks”
- Setup:

```
. matrix A = I(4)
. matrix B1 = (.,0,0,0 \ .,.,0,0 \ .,.,.,0 \ .,.,.,.)
. matrix B2 = (.,.,0,0 \ .,.,.,0 \ .,.,.,0 \ .,0,0,.)
. matlist B2
```

| | c1 | c2 | c3 | c4 |
|----|----|----|----|----|
| r1 | . | . | 0 | 0 |
| r2 | . | . | . | 0 |
| r3 | . | . | . | 0 |
| r4 | . | 0 | 0 | . |

- Estimation:

SVAR-IV estimation of oil shock IRFs II


```
. // (constraint setup omitted)
. // get starting values using orthogonalized B matrix
. quietly svar inflation ogap fedfunds netoil, aeq(A) beq(B1) ///
>     lags(1/8) varconstraint(1/12)
.
. // estimation using the instrument-implied B matrix
. matrix b = e(b)
. quietly svar inflation ogap fedfunds netoil, aeq(A) beq(B2) ///
>     lags(1/8) from(b) noidencheck varconstraint(1/12)
.
. // parameter estimates; column 1 is identified
. matlist e(B)
```

| | inflation | ogap | fedfunds | netoil |
|-----------|-----------|----------|-----------|----------|
| inflation | .4824677 | .1998924 | 0 | 0 |
| ogap | -.2000096 | .614898 | -.3432701 | 0 |
| fedfunds | .0123024 | .5273174 | .505104 | 0 |
| netoil | 2.137743 | 0 | 0 | 6.310632 |

- Graphs:

```
. irf graph sirf, irf(svar_iv) impulse(inflation) ///
>     yline(0) xlabel(0(4)20) byopts(yrescale)
```

SVAR-IV estimation of oil shock IRFs III



Graphs by irfname, impulse variable, and response variable

Summary

- Instruments aid in solving the IRF identification problem.
- SVAR-IVs can be specified as large SVAR models with particular restrictions imposed by the nature of the instruments.
- Thus some SVAR-IVs can be estimated using existing svar tools.
- I showed SVAR-IV estimation of the effects of net oil price shocks.
- Extensions possible combining IV and LP, as well.

Thank you!